À la recherche d'une relation de récurrence

Le but ici est de trouver une relation de récurrence entre des primitives des fonctions

$$f_n : x \mapsto \frac{1}{(1+x^2)^n} \quad (n \in \mathbf{N})$$

```
> load("ipp.mc")$
> f[n](x) := 1/(x^2+1)^n$
> I[n](x) := integrate(f[n](x),x)$
> A:Ipp(1,f[n](x),x);
```

$$\int \frac{1}{\left(x^2+1\right)^n}\,\mathrm{d}x = 2\,n\int x^2\,\left(x^2+1\right)^{-n-1}\,\mathrm{d}x + \frac{x}{\left(x^2+1\right)^n}$$

```
> C:block(B:'(f[n](x)-f[n+1](x)),B = ev(B,radcan,nouns));
```

$$f_n(x) - f_{n+1}(x) = x^2\,\left(x^2+1\right)^{-n-1}$$

```
> D:'(I[n](x)-I[n+1](x)) = integrate(rhs(C),x);
```

$$I_n(x) - I_{n+1}(x) = \int x^2\,\left(x^2+1\right)^{-n-1}\,\mathrm{d}x$$

```
> E:subst(lhs(D),rhs(D),subst('(I[n](x)),I[n](x),A));
```

$$I_n(x) = 2\,n\,(I_n(x) - I_{n+1}(x)) + \frac{x}{\left(x^2+1\right)^n}$$

```
> F:solve(E,'(I[n+1](x)));
```

$$\left[I_{n+1}(x) = \frac{(2\,n-1)\,\left(x^2+1\right)^n\,I_n(x) + x}{2\,n\,\left(x^2+1\right)^n}\right]$$

```
> distrib(part(F,1));
```

$$I_{n+1}(x) = \frac{(2\,n-1)\,I_n(x)}{2\,n} + \frac{x}{2\,n\,\left(x^2+1\right)^n}$$

Voici la relation de récurrence recherchée. On peut donc, de proche en proche, calculer les primitives des fonction f_n... Ce que MAXIMA sait faire directement.

```
> load("integration.mc")$
> primitive(f[2](x),x);
```

$$\int \frac{1}{\left(x^2+1\right)^2}\,\mathrm{d}x = \frac{\operatorname{Arctan} x}{2} + \frac{x}{2\,x^2+2}$$

```
> primitive(f[4](x),x);
```

$$\int \frac{1}{\left(x^2+1\right)^4}\, dx = \frac{5\,\mathrm{Arctan}\, x}{16} + \frac{15\,x^5+40\,x^3+33\,x}{48\,x^6+144\,x^4+144\,x^2+48}$$

```
> primitive(f[5](x),x);
```

$$\int \frac{1}{\left(x^2+1\right)^5}\, dx = \frac{35\,\mathrm{Arctan}\, x}{128} + \frac{105\,x^7+385\,x^5+511\,x^3+279\,x}{384\,x^8+1536\,x^6+2304\,x^4+1536\,x^2+384}$$